
Vlastní procedury a funkce

Dnes si ukážeme, jak pro opakující se akce můžeme vytvořit vlastní procedury (pro výpočty – vlastní funkce). Zatím tyto procedury a funkce budeme vytvářet pouze v našem formuláři, příští přednášku si ukážeme kam je umístit, abychom je mohli používat i v jiných programech.

Programování ve Visual Basicu je, jak jsme si již řekli, procedurálním programováním. My vlastně neustále programujeme jen krátké „podprogramy“ tzv. událostní procedury, které zpracovávají události vyvolané uživatelem nebo systémem.

Proč vytvářet vlastní procedury?

Většinou pomocí procedur, řešíme opakující se část programu nebo ty části programu, které se od sebe jen nepatrně liší (různými hodnotami parametrů).  Často si může uživatel vybrat, jak bude vykonávat danou akci (klepnutím na tlačítko), výběrem z nabídky případně se tato akce vykonává ihned po spuštění programu. Je proto vhodnější než na několika místech programu psát opakující se řádky, raději vytvořit jednu proceduru, která tuto akci vykonává a pak ji jen na různých místech programu (z různých událostních procedur volat).

Kreslení bodů na formuláři (Body/Body1.vbp)

Trošičku umělý příklad, který na formuláři kreslí náhodně umístěné body. Kreslení bodů vyvoláváme pomocí dvou tlačítek a nabídkou.

Procedura, která na formulář vykreslí body. Tato procedura má dva parametry (viz. níže), parametr Počet, který udává kolik bodů na formulář kreslím a parametr barva určující barvu bodů

Vlastní procedury píši v části General Declaration uvozuji je klíčovým slovem Sub a jménem, konec End Sub je doplněn automaticky. U parametrů procedury určím jakého typu parametry budou.

Sub Kresli(Pocet As Integer, barva As Long)

'na formular vykresli Pocet bodu barvy urcene parametrem barva

'pozice bodu nahodne generovana

Dim i As Integer

 i = 1

Do While i <= Pocet

    'kreslim

    Form1.PSet (Rnd() * Form1.ScaleWidth, Rnd() * Form1.ScaleHeight), barva

    i = i + 1

Loop

End Sub

Jak proceduru voláme?

Kresli 10, vbRed

nebo

N = 5

b = vbBlue

Kresli N, b

‘N a b jsou globální proměnné

Zatím jsme si vůbec nic neřekli o parametrech, které do procedur vkládáme

Parametry procedur (Body/Body2.vbp)

Důležitou vlastností všech procedur a funkcí jsou parametry, pomocí níž můžete do funkce předat nějaké hodnoty nebo je z funkce získat. Je dobrým zvykem vytvářet procedury tak, aby tvořily „uzavřený“ celek, do kterého se přistupuje pouze pomocí parametrů. Parametry se užívají v těle procedury, kde svými identifikátory zastupují skutečné hodnoty, které je při konkrétním volání funkce nahradí. 

Rozlišujeme parametry dvojího druhu:

1.  Parametry volané odkazem (vstupně-výstupní). Toto volání je ve VisualBasicu standardní. Do funkce se předá pouze odkaz na příslušnou proměnnou. Znamená to tedy, že každá změna, kterou procedura provede do parametru se promítá i do vnější proměnné. Tento typ volání užíváme pro předání výsledku z procedury do volajícího programu, dále tento způsob volání použijeme tehdy, když je předávaný parametr příliš velký (co do velikosti paměti, např. rozsáhlé pole). Pokud chceme deklarovat takovýto parametr můžeme použít klíčové slovo ByRef.

Pozor Visual Basic si umí “poradit“, pokud mu do parametru volaného odkazem vložím pouze hodnotu a ne proměnnou.

2.  Parametry volané hodnotou (vstupní). Skutečná hodnota proměnné se „zkopíruje“ na jiné místo v paměti a procedura potom pracuje s touto „kopií“. Proto veškeré změny, které v parametru během provádění procedury provedete se do proměnné v hlavním programu nepromítnou. Tento způsob předávání parametrů má výhodu v tom, že hodnota proměnné hlavního programu nebude změněna. Můžete tedy s parametrem v proceduře zacházet zcela libovolně. Nevýhodou je nutnost přenosu hodnoty, což zvyšuje v konečném důsledku nároky na paměť. Definici toho typu parametru ve Visual Basicu předchází klíčové slovo ByVal.
Pozor pokud tedy neurčíme jinak jsou parametry předávané odkazem. Proto pokud bychom chtěli programovat „čistě“, měli bychom vždy do takového parametru předávat proměnnou a ne jen hodnotu. Pomocí těchto parametrů, získáváme i hodnoty z procedury nazpět. Malé nebezpečí spočívá v tom, že procedura tedy může měnit hodnoty globálních proměnných. Jako ukázku si ukážeme stejný příklad, kde ale změníme hodnotu parametru Počet.

Sub Kresli(Pocet As Integer, barva As Long)

'na formular vykresli Pocet bodu barvy urcene parametrem barva

'pozice bodu nahodne generovana

Do While Pocet > 0

    'kreslim

    Form1.PSet (Rnd() * Form1.ScaleWidth, Rnd() * Form1.ScaleHeight), barva

    Pocet = Pocet - 1

Loop

End Sub

Tentokrát body kreslíme tak, že postupně zmenšujeme hodnotu proměnné Počet.

Proceduru pak zavoláme takto a po skončení procedury necháme zobrazit MsgBox se správou kolik bodů jsme nakreslili.

N = 5

Kresli N, vbGreen

MsgBox "Na obrazovce je " & N & " nakreslených bodů"

A podívejte se, co se nám vypíše: Globální proměnná je procedurou pomocí parametru Počet volaného odkazem přepsána na hodnotu 0.

Naše procedura by neměla měnit hodnoty žádného parametrů, použité parametry by bylo vhodnější, vzhledem k jejich použití a i možnostem volání procedury, předávat hodnotou. Tedy takto:

Sub Kresli(ByVal Pocet As Integer, ByVal barva As Long)

Pokud teď použijeme stejné volání, tak počet zobrazovaných bodů bude již správný. Procedura si vytvořila dvě nové lokální proměnné, do kterých hodnoty z parametrů překopírovala.

Funkce ve Visual Basicu

Dnes si ukážeme jak vytvářet další z vlastních podprogramů a to konkrétně funkci. Funkce  je velmi podobná proceduře, rozdíl spočívá v tom, že funkce vrací návratovou hodnotu. Funkce provede výpočty se zadanými parametry a vrátí výsledek.Většinou tedy její volání stojí na pravé straně nějakého přiřazovacího příkazu. Protože funkce vrací výsledek musím definovat také její typ. Vytvořím jí opět v části General Declaration.

Function Secti(A As Integer, B As Integer) As Integer


Secti = A+B

End Function

volání:

P = Secti(10,24)

Již jsme si ukazovali, jak vypočítat největšího společného dělitele.

Pojďme si tento příklad předělat pomocí funkce. (NSD/NSDf.vbp)

Function NSD(ByVal A As Integer, B As Integer) As Integer

'vypocte nejvetsiho spolecneho delitele cisel A a B

'eukliduv algoritmus

Do While A <> B

    If A > B Then

        A = A - B

    Else

        B = B - A

    End If

Loop

NSD = A

End Function

a její volání (včetně použití InputBoxu)

Dim C1 As Integer, C2 As Integer 'cisla jejichz NSD budu urcovat

C1 = Val(InputBox("Zadej A:", "NSD", 18)) 'uzivatel zada A, nastaveno 12

C2 = Val(InputBox("Zadej B:", "NSD", 12)) 'uzivatel zada B, nastaveno na 6

cbNSD.Visible = False

'vypocet NSD rovnou soucasti zaverezneho zobrazeni

lbVysl.Caption = "NSD " & C1 & " a " & C2 & " je " & NSD(C1, C2)
lbVysl.Visible = True

Zkusme vytvořit funkci, která sečte aritmetickou řadu čísel. Funkce bude mít opět dva parametry první člen řady, poslední člen řady. V prvním přiblížení předpokládejme, že se jedná o řadu jejíž diference (rozdíl mezi sousedními členy) je rovna 1.

Function Soucet_Rady(ByVal Pocatek As Integer, ByVal Konec As Integer) As Integer

'secte radu cisel s diferenci 1, rada zacina cislem ulozenym v parametru Pocatek _

a konci cislem ulozenym v parametru Konec

    Dim i As Integer

    Dim S As Integer

    S = 0

    For i = Pocatek To Konec

       S = S + i

    Next i

    Soucet_Rady = S

End Function

K funkcím jejich parametrům a typům se ještě podrobně vrátíme, teď malinko odbočíme a řekneme si k čemu je modul a jak ho ve Visual Basicu využijeme.

Modul

Už Vás asi napadlo a to vždy když budu potřebovat použít nějakou funkci a nejen funkci, kterou jsme již někdy naprogramovali si ji budeme muset znovu ve svém novém projektu vytvářet? Můžeme ji sice okopírovat, ale asi by bylo efektivnější kdybychom ve svých programech mohli mít něco jako „knihovnu“ funkcí a tu si ke svým projektům přidávat, tak jak budeme potřebovat.

Takové knihovny jsou opravdu již připravené, neříká se jim knihovny, ale moduly a přidáme si je do projektu následujícím způsobem, z nabídky Project vybereme položku Add Modul, zde si vybere buď úplně nový prázdný modul nebo již modul existující.

Modul nemá žádnou vnější podobu je to jen kód obsahující procedury a funkce, které lze používat ve všech formulářích mého projektu. 

Přesuneme naši funkci do tohoto modulu, v kódu programu nemusím nic měnit, automaticky se využívá funkce z modulu. Pokud si modul uložím mohu ho lehce přidat některému jinému projektu a tam naprogramované funkce opětovně využívat.

VBA (Visual Basic for Application)

No a teď hurá do Excelu. Začneme trošičku netradičně, a to vytvářením vlastních funkcí v Excelu. Vlastně vezmeme naše funkce z Visual Basicu do Excelu  a budeme je používat tam.

Máme otevřený Excel, chceme si vytvořit naší novou vlastní funkci. Začneme nějakou jednoduchou, kterou již dobře umíme, co takhle náš oblíbený NSD (největší společný dělitel).

Do Visual Basicu vExcelu se přesuneme buď kombinací kláves Alt + F11, nebo výběrem položky Makro v nabídce Nástroje. dříve se „programům“ v Excelu říkalo makra, tento název se vžil a tak ho také budeme používat, rovnou se přesuneme do Editoru jazyka VisualBasic.

Prostředí by Vám silně mělo připomínat náš oblíbený jazyk VisualBasic, nenechte se zmást, že je jinak uspořádané – lze to změnit. 

Jak vypadá projekt v Excelu. Základním prvkem je tu zpracovávaný sešit s jeho listy, nechte si otevřený jen ten, který potřebujete. My se nebudeme zabývat zpracováváím jednotlivých listů, nastavováním jejich vlastností apod. ne že by to nešlo, ale není na to prostor ani čas.

První, co se chceme naučit je, jak udělat vlastní funkci. Měla by to být funkce dostupná ze všech listů daného sešitu, ideálně ještě přenositelná do jiných sešitů. Když jsme před chvílí chtěli psát univerzální funkce ve Visual Basicu, kam jsme je uložili?

Modul – vytvoření

ano, do modulu. Modul v Excelu se nám buď vytvoří sám, to když nahráváme makra (příště) nebo ho vytvoříme sami, velmi podobně jako ve Visual Basicu. Zde je to Insert Modul.

Napíši klíčové slovo Function a tvořím.

Pozor nějaké rozdíly tu jsou. Tento modul nelze samostatně uložit, ukládá se vždy s celým sešitem. Lze ho ale exportovat a importovat. Také lze přetáhnout třeba již vytvořené moduly z Visual Basicu.

No a jak funkci voláme? Kromě jiného ji můžeme používat přímo v Excelu, úplně stejně jako kdyby to byla původní funkce Excelu. Je zařazena do kategorie vlastní  funkce.

Dnes si ukážeme několik funkcí, na kterých si vysvětlíme jaké další parametry funkcím můžeme předávat a jak se s nimi pracuje. 

Prvočíslo

Boolovská funkce, dostane číslo a určí zda je toto číslo prvočíslem.

pořádně si rozmyslete kterými všemu čísly má smysl zkoušet dělit parametr A, abychom určili zda se jedná o prvočíslo.

Function Prvocislo(A As Integer) As Boolean

Dim I As Integer

I = 2

Prvocislo = True

Do While I <= Sqr(A) And Prvocislo

    If (A Mod I) = 0 Then Prvocislo = False

    I = I + 1

Loop

End Function

Nepovinné parametry Optional

Sčítání řady

Funkce sečte členy aritmetické řady. Řada je zadána prvním prvkem, posledním prvkem a diferencí mezi prvky (parametr Krok).

Function Soucet_Rady(Pocatek As Integer, Konec As Integer, Optional Krok = 1) As Integer

'soucet clenu aritmeticke rady, Pocatek - prvni clen rady, Konec - posledni clen rady

'Krok - diference aritmeticke rady

    Dim i As Integer

    Dim S As Integer

    S = 0

    For i = Pocatek To Konec Step Krok

       S = S + i

    Next i

    Soucet_Rady = S

End Function

Možná jste si všimli dvou novinek v naší funkci. První menší novinka je použití For cyklu i s jiným přírůstkem než 1, klíčové slovo Step.

Druhou novinkou je použití tzv. používání nepovinných parametrů. Tyto nepovinné parametry poznáte podle toho, že je před jejich jménem použito klíčové slovo Optional. Po té, co jeden z parametrů deklaruji jako nepovinný, musí být všechny následující parametry deklarovány jako nepovinné.

Naší funkcí Soucet_Rady mohu nyní volat se dvěma nebo třemi parametry. Je výhodné jak je naznačeno v příkladu používat tento nepovinný parametr s výchozí hodnotou. V případě, že je funkce volána bez hodnoty nepovinného parametru, použije funkce namísto tohoto parametru jeho výchozí hodnotu.

Výskyt

Vytvořte funkci, která spočte kolikrát se daný řetězec vyskytuje v jiném řetězci.

Function Vyskyt(ByVal Co As String, ByVal Kde As String) As Integer

'spocte kolikrát se daný retizec Co  vyskytuje v jiném retezci Kde

Dim I As Integer

Dim Start As Integer

Dim Pocet As Integer

Pocet = 0

I = 1

Start = 1

Do While Start < Len(Kde)

 I = InStr(Start, Kde, Co)

 If I > 0 Then

    Start = I + 1

    Pocet = Pocet + 1

Else: Start = Len(Kde)

 End If

Loop

Vyskyt = Pocet

End Function

Vyzkoušejte si nejprve naprogramování funkce, která vůbec určí zda tam daný řetězec je. Zkuste si i tuto funkci naprogramovat a nepoužít funkci vestavěnou.

Narozeniny

Vytvořte funkci, která z data narození zjistí, zda dotyčná osoba již letos měla narozeniny.

Function Mel_Naroz(ByVal Datum As Date) As Boolean

'urci z datumu narozeni, zda jiz v danem roce narozeniny byly

Mel_Naroz = False   'predpokladam, ze nebyly

'pokud plati nasledujici nerovnosti tak byly

If Month(Datum) < Month(Date) Then

    Mel_Naroz = True

ElseIf Month(Datum) = Month(Date) Then

    If Day(Datum) < Day(Date) Then

        Mel_Naroz = True

    End If

End If

End Function

Objekt Range

Součet hodnot z vybrané oblasti (SUMA)

Práce s parametrem „oblast buněk“. Použijeme speciální datový typ Range (jedná se o proměnnou typu Object, platí pro ni vše, co jsme si ukazovali minule). 

Objekt Range bude hlavním objektem, kterým se ve VBA budeme zabývat. Pomocí něj a jeho vlastností se odkazujeme na jednotlivé buňky Excelu. Dnes si ukážeme jen několik základních funkcí pro práci s tímto objektem. V Příští přednášce si tyto poznatky ještě rozšíříme.

Function Suma(Oblast As Range) As Double

'secte hodnoty bunek ve vybrane oblasti

Dim S As Double

Dim o As Range

S = 0

For Each o In Oblast

    S = S + o.Value

Next o

Suma = S

End Function

V příkladě jsme poznali první vlastnost objektu Range, a tou je vlastnost Value  - v případě jedné buňky je to její hodnota.

IsEmpty

Aritmetický průměr

Pro práci s objekty se nám hodí speciální fce IsEmpty, pomůže nám rozlišit buňky s nulovou hodnotou od prázdných. Funkce vrazí hodnotu TRUE, pokud proměnná dosud nebyla inicializovaná. V našem případě určí zda je buňka prázdná.

Function Muj_Prumer(Oblast As Range) As Double

'vypocte aritmeticky prumer ze vsech bunek zadane oblasti

Dim S As Double

Dim Pocet As Integer

S = 0

Pocet = 0

For Each o In Oblast

If Not (IsEmpty(o)) Then

    S = S + o.Value

    Pocet = Pocet + 1

End If

Next o

If Pocet > 0 Then

    Muj_Prumer = S / Pocet

Else

    Muj_Prumer = 0

End If

End Function

Další podrobnosti o objektu Range naleznete v nápovědě.

Použití vestavěnných funkcí

Určitě Vás již napadlo, zda bych ve svých funkcích nemohla použít vestavěné funkce Excelu, tak jak je známe ze zimního semestru. Můžeme, ale musíme použít jejich anglické názvy a tuto syntaxi.

WorkSheetFunction.anglické jméno funkce například nalezení Minima z oblasti buněk

WorkSheetFunction.Min

Vytvoř funkci, která porovná zadanou hodnotou s minimem z množiny hodnot a vrátí -1 pokud je zadaná hodnota menší, 0 pokud je rovna a 1 pokud je větší.

Function Porovnej(Co As Double, Oblast As Range) As Integer

'porovná zadanou hodnotou Co s minimem z množiny hodnot Oblast

'a vrátí -1 pokud je zadaná hodnota menší, 0 pokud je rovna a 1 pokud je vìtší nez minimum

Dim m As Double

m = WorksheetFunction.Min(Oblast)

If Co > m Then

    Porovnej = 1

ElseIf Co = m Then

    Porovnej = 0

Else

    Porovnej = -1

End If

End Function

JE TAM

Vytvořte funkci, která zjistí zda se hledaná hodnota nachází ve vybrané oblasti buněk.

Function JeTam(Co As Variant, Kde As Range) As Boolean

'funkce zjisti zda se hodnota Co nachazi ve vybrane oblasti bunek Kde

Dim o As Range

JeTam = False

For Each o In Kde

 If o.Value = Co Then

    JeTam = True

    Exit For

 End If

Next o

End Function
































